
PART I

Joint work with Michael Hutter



The first year of primary school

Available numbers (digits): (0), 1, 2, 3, 4, 5, 6, 7, 8, 9

Addition
3 + 5 = ?
2 + 7 = ?
4 + 3 = ?

Subtraction
7− 5 = ?
5− 1 = ?
9− 3 = ?

I All results are in the set of available numbers
I No confusion for first-year school kids

Programming today

Available numbers: 0, 1, . . . , 255

Addition
uint8_t a = 42;
uint8_t b = 89;
uint8_t r = a + b;

Subtraction
uint8_t a = 157;
uint8_t b = 23;
uint8_t r = a - b;

I All results are in the set of available numbers
I Larger set of available numbers: uint16_t, uint32_t, uint64_t
I Basic principle is the same; for the moment stick with uint8_t

Still in the first year of primary school
Crossing the ten barrier

6 + 5 = ?
9 + 7 = ?
4 + 8 = ?

I Inputs to addition are still from the set of available numbers
I Results are allowed to be larger than 9

I Addition is allowed to produce a carry

What happens with the carry?
I Introduce the decimal positional system
I Write an integer A in two digits a1a0 with

A = 10 · a1 + ·a0

I Note that at the moment a1 ∈ {0, 1}

. . . back to programming

uint8_t a = 184;
uint8_t b = 203;
uint8_t r = a + b;

I The result r now has the value of 131

I The carry is lost, what do we do?
I Could cast to uint16_t, uint32_t etc.,

but that solves the problem only for this uint8_t example
I We really want to obtain the carry, and put it into another uint8_t

The AVR ATmega

I 8-bit RISC architecture
I 32 registers R0. . . R31, some of those are “special”:

I (R26,R27) aliased as X
I (R28,R29) aliased as Y
I (R30,R31) aliased as Z
I X, Y, Z are used for addressing
I 2-byte output of a multiplication always in R0,R1

I Most arithmetic instructions cost 1 cycle
I Multiplication and memory access takes 2 cycles

Multiprecision arithmetic 7



184 + 203

LDI R5, 184
LDI R6, 203
ADD R5, R6 ; result in R5, sets carry flag
CLR R6 ; set R6 to zero
ADC R6,R6 ; add with carry, R6 now holds the carry

Multiprecision arithmetic 8



Later in primary school

Addition
42 + 78 = ?
789 + 543 = ?
7862 + 5275 = ?

7862
+ 5275
+ 13137

I Once school kids can add
beyond 1000, they can add
arbitrary numbers

Multiprecision addition is old

“Oh L̄ılāvat̄ı, intelligent girl, if you understand addition and
subtraction, tell me the sum of the amounts 2, 5, 32, 193, 18,
10, and 100, as well as [the remainder of] those when
subtracted from 10000.”

—“L̄ılāvat̄ı” by Bhāskara (1150)

Multiprecision arithmetic 10



Why do we do all that again?

I Mainly asymmetric cryptography heavily relies on multiprecision
arithmetic

I Example 1: RSA-2048 needs (modular) multiplication and squaring
of 2048-bit numbers

I Example 2:
I Elliptic curves defined over finite fields
I Typically use EC over large-characteristic prime fields
I Typical field sizes: (160 bits, 192 bits), 256 bits, 414 bits . . .

I Example 3:
I Genus-2 hyperelliptic curves defined over finite fields
I Typically use HEC over large-characteristic prime fields
I Field size for 128-bit security: 128 bits

I For now mainly interested in 160-bit and 256-bit arithmetic

AVR multiprecision addition. . .

I Add two n-byte numbers, returning an n + 1 byte result:
I Input pointers X,Y, output pointer Z

LD R5,X+
LD R6,Y+
ADD R5,R6
ST Z+,R5

LD R5,X+
LD R6,Y+
ADC R5,R6
ST Z+,R5

LD R5,X+
LD R6,Y+
ADC R5,R6
ST Z+,R5

LD R5,X+
LD R6,Y+
ADC R5,R6
ST Z+,R5

...

CLR R5
ADC R5,R5
ST Z+,R5

. . . and subtraction

I Add two n-byte numbers, returning an n + 1 byte result:
I Input pointers X,Y, output pointer Z
I Use highest byte = −1 to indicate negative result

LD R5,X+
LD R6,Y+
SUB R5,R6
ST Z+,R5

LD R5,X+
LD R6,Y+
SBC R5,R6
ST Z+,R5

LD R5,X+
LD R6,Y+
SBC R5,R6
ST Z+,R5

LD R5,X+
LD R6,Y+
SBC R5,R6
ST Z+,R5

...

CLR R5
SBC R5,R5
ST Z+,R5

Multiprecision arithmetic 13



How about multiplication?

I Consider multiplication of 1234 by 789

1234 · 789
11106
98726

+ 863806
973626

I This is also an old technique
I Earliest reference I could find is again the L̄ılāvat̄ı (1150)

How about multiplication?

I Consider multiplication of 1234 by 789

1234 · 789
+ 11106

666
666
666

I This is also an old technique
I Earliest reference I could find is again the L̄ılāvat̄ı (1150)

How about multiplication?

I Consider multiplication of 1234 by 789

+ 1234 · 789
20978

666
666
666

I This is also an old technique
I Earliest reference I could find is again the L̄ılāvat̄ı (1150)

1234 · 789
20978

Let’s do that on the AVR

LD R2, X+
LD R3, X+
LD R4, X+

LD R7, Y+

MUL R2,R7
ST Z+,R0
MOV R8,R1

MUL R3,R7
ADD R8,R0
CLR R9
ADC R9,R1

MUL R4,R7
ADD R9,R0
CLR R10
ADC R10,R1

LD R7, Y+

MUL R2,R7
MOVW R12,R0

MUL R3,R7
ADD R13,R0
CLR R14
ADC R14,R1

MUL R4,R7
ADD R14,R0
CLR R15
ADC R15,R1

ADD R8,R12
ST Z+,R8
ADC R9,R13
ADC R10,R14
CLR R11
ADC R11,R15

LD R7, Y+

MUL R2,R7
MOVW R12,R0

MUL R3,R7
ADD R13,R0
CLR R14
ADC R14,R1

MUL R4,R7
ADD R14,R0
CLR R15
ADC R15,R1

ADC R9,R12
ST Z+,R9
ADC R10,R13
ADC R11,R14
CLR R12
ADC R12,R15

ST Z+,R10
ST Z+,R11
ST Z+,R12

Multiprecision arithmetic 15



Let’s do that on the AVR

LD R2, X+
LD R3, X+
LD R4, X+

LD R7, Y+

MUL R2,R7
ST Z+,R0
MOV R8,R1

MUL R3,R7
ADD R8,R0
CLR R9
ADC R9,R1

MUL R4,R7
ADD R9,R0
CLR R10
ADC R10,R1

LD R7, Y+

MUL R2,R7
MOVW R12,R0

MUL R3,R7
ADD R13,R0
CLR R14
ADC R14,R1

MUL R4,R7
ADD R14,R0
CLR R15
ADC R15,R1

ADD R8,R12
ST Z+,R8
ADC R9,R13
ADC R10,R14
CLR R11
ADC R11,R15

LD R7, Y+

MUL R2,R7
MOVW R12,R0

MUL R3,R7
ADD R13,R0
CLR R14
ADC R14,R1

MUL R4,R7
ADD R14,R0
CLR R15
ADC R15,R1

ADC R9,R12
ST Z+,R9
ADC R10,R13
ADC R11,R14
CLR R12
ADC R12,R15

ST Z+,R10
ST Z+,R11
ST Z+,R12

Multiprecision arithmetic 15



Let’s do that on the AVR

LD R2, X+
LD R3, X+
LD R4, X+

LD R7, Y+

MUL R2,R7
ST Z+,R0
MOV R8,R1

MUL R3,R7
ADD R8,R0
CLR R9
ADC R9,R1

MUL R4,R7
ADD R9,R0
CLR R10
ADC R10,R1

LD R7, Y+

MUL R2,R7
MOVW R12,R0

MUL R3,R7
ADD R13,R0
CLR R14
ADC R14,R1

MUL R4,R7
ADD R14,R0
CLR R15
ADC R15,R1

ADD R8,R12
ST Z+,R8
ADC R9,R13
ADC R10,R14
CLR R11
ADC R11,R15

LD R7, Y+

MUL R2,R7
MOVW R12,R0

MUL R3,R7
ADD R13,R0
CLR R14
ADC R14,R1

MUL R4,R7
ADD R14,R0
CLR R15
ADC R15,R1

ADC R9,R12
ST Z+,R9
ADC R10,R13
ADC R11,R14
CLR R12
ADC R12,R15

ST Z+,R10
ST Z+,R11
ST Z+,R12

Multiprecision arithmetic 15



Let’s do that on the AVR

LD R2, X+
LD R3, X+
LD R4, X+

LD R7, Y+

MUL R2,R7
ST Z+,R0
MOV R8,R1

MUL R3,R7
ADD R8,R0
CLR R9
ADC R9,R1

MUL R4,R7
ADD R9,R0
CLR R10
ADC R10,R1

LD R7, Y+

MUL R2,R7
MOVW R12,R0

MUL R3,R7
ADD R13,R0
CLR R14
ADC R14,R1

MUL R4,R7
ADD R14,R0
CLR R15
ADC R15,R1

ADD R8,R12
ST Z+,R8
ADC R9,R13
ADC R10,R14
CLR R11
ADC R11,R15

LD R7, Y+

MUL R2,R7
MOVW R12,R0

MUL R3,R7
ADD R13,R0
CLR R14
ADC R14,R1

MUL R4,R7
ADD R14,R0
CLR R15
ADC R15,R1

ADC R9,R12
ST Z+,R9
ADC R10,R13
ADC R11,R14
CLR R12
ADC R12,R15

ST Z+,R10
ST Z+,R11
ST Z+,R12

Multiprecision arithmetic 15



Let’s do that on the AVR

I Problem: Need 3n + c registers for n×n-byte multiplication

I Can add on the fly, get down to 2n + c, but more carry handling

Can we do better?

“Again as the information is understood, the multiplication of
2345 by 6789 is proposed; therefore the numbers are written
down; the 5 is multiplied by the 9, there will be 45; the 5 is put,
the 4 is kept; and the 5 is multiplied by the 8, and the 9 by the
4 and the products are added to the kept 4; there will be 80;
the 0 is put and the 8 is kept; and the 5 is multiplied by the 7
and the 9 by the 2 and the 4 by the 8, and the products are
added to the kept 8; there will be 102; the 2 is put and the 10
is kept in hand. . . ”

From “Fibonacci’s Liber Abaci”, Chapter 2 (English translation by Sigler)

Product scanning on the AVR

LD R2, X+
LD R3, X+
LD R4, X+
LD R7, Y+
LD R8, Y+
LD R9, Y+

MUL R2, R7
MOV R13, R1
STD Z+0, R0
CLR R14
CLR R15

MUL R2, R8
ADD R13, R0
ADC R14, R1
MUL R3, R7
ADD R13, R0
ADC R14, R1
ADC R15, R5
STD Z+1, R13
CLR R16

MUL R2, R9
ADD R14, R0
ADC R15, R1
ADC R16, R5
MUL R3, R8
ADD R14, R0
ADC R15, R1
ADC R16, R5
MUL R4, R7
ADD R14, R0
ADC R15, R1
ADC R16, R5
STD Z+2, R14
CLR R17

MUL R3, R9
ADD R15, R0
ADC R16, R1
ADC R17, R5
MUL R4, R8
ADD R15, R0
ADC R16, R1
ADC R17, R5
STD Z+3, R15

MUL R4, R9
ADD R16, R0
ADC R17, R1
STD Z+4, R16

STD Z+5, R17

Even better. . . ?

From the Treviso Arithmetic, 1478
(http://www.republicaveneta.com/doc/abaco.pdf)

Hybrid multiplication

I Idea: Chop whole multiplication into smaller blocks
I Compute each of the smaller multiplications by schoolbook
I Later add up to the full result
I See it as two nested loops:

I Inner loop performs operand scanning
I Outer loop performs product scanning

I Originally proposed by Gura, Patel, Wander, Eberle, Chang Shantz,
2004

I Various improvements, consider 160-bit multiplication:
I Originally: 3106 cycles
I Uhsadel, Poschmann, Paar (2007): 2881 cycles
I Scott, Szczechowiak (2007): 2651 cycles
I Kargl, Pyka, Seuschek (2008): 2593 cycles

Operand-caching multiplication

I Hutter, Wenger, 2011: More efficient way to decompose
multiplication

I Inside separate chunks use product-scanning
I Main idea: re-use values in registers for longer

I Performance:
I 2393 cycles for 160-bit multiplication
I 6121 cycles for 256-bit multiplication

I Followup-paper by Seo and Kim: “Consecutive operand caching”:
I 2341 cycles for 160-bit multiplication
I 6115 cycles for 256-bit multiplication

I Followup-paper by Seo and Kim: “Consecutive operand caching”:
I 2341 cycles for 160-bit multiplication
I 6115 cycles for 256-bit multiplication

Multiplication complexity

I So far, multiplication of 2 n-byte numbers needs n2 MULs
I Kolmogorov conjectured 1952: You can’t do better, multiplication

has quadratic complexity

I Proven wrong by 23-year old student Karatsuba in 1960
I Idea: write A ·B as (A0 + 2mA1)(B0 + 2mB1) for half-size

A0, B0, A1, B1

I Compute

A0B0 + 2m(A0B1 + B0A1) + 22mA1B1

=A0B0 + 2m((A0 + A1)(B0 + B1)−A0B0 −A1B1) + 22mA1B1

I Recursive application yields Θ(nlog2 3) runtime

Does that help on the AVR?

Multiprecision arithmetic 22



The straight-forward approach

Consider multiplication of n-byte numbers

A =̂ (a0, . . . , an−1) and
B =̂ (b0, . . . , bn−1)

I Write A = A` + 28kAh and B = B` + 28kBh

for k-byte integers A`, Ah, B`, and Bh and k = n/2

I Compute L = A` ·B` =̂ (`0, . . . , `n−1)

I Compute H = Ah ·Bh =̂ (h0, . . . , hn−1)

I Compute M = (A` + Ah) · (B` + Bh) =̂ (m0, . . . ,mn)

I Obtain result as A ·B = L + 28k(M − L−H) + 28nH

Multiplication by the carry in M

I Can expand carry to 0xff or 0x00
I Use AND instruction for multiplication

I Does not help for recursive Karatsuba

Subtractive Karatsuba
I Compute L = A` ·B` =̂ (`0, . . . , `n−1)

I Compute H = Ah ·Bh =̂ (h0, . . . , hn−1)

I Compute M = |A` −Ah| · |B` −Bh| =̂ (m0, . . . ,mn−1)

I Set t = 0, if M = (A` −Ah) · (B` −Bh); t = 1 otherwise
I Compute M̂ = (−1)tM = (A` −Ah)(B` −Bh)

=̂ (m̂0, . . . , m̂n−1)

I Obtain result as A ·B = L + 28k(L + H − M̂) + 28nH

Conditional negation

The easy solution

if(b) a = -a

I NEG instruction does not help for multiprecision
I Can subtract from zero, but subtraction would overwrite zero
I Even worse, the if would create a timing side-channel!

The constant-time solution
I Produce condition bit as byte 0xff or 0x00
I XOR all values with this condition byte

I Negate the condition byte and obtain 0x01 or 0x00
I Add this value to the lowest byte
I Ripple through the carry (ADC with zero)

Conditional negation

The easy solution

if(b) a = -a

I NEG instruction does not help for multiprecision
I Can subtract from zero, but subtraction would overwrite zero
I Even worse, the if would create a timing side-channel!

The constant-time solution
I Produce condition bit as byte 0xff or 0x00
I XOR all values with this condition byte
I Don’t negate the condition byte
I Subtract the condition byte (0xff or 0x00 from all bytes)
I Saves two NEG instructions

Refined Karatsuba

I Consider example of 4×4-byte Karatsuba multiplication:

l0 l1 l2 l3 h0 h1 h2 h3

- m̂0 m̂1 m̂2 m̂3

+ l0 l1 l2 l3
+ h0 h1 h2 h3

I Karatsuba performs some additions twice
I Refined Karatsuba: do them only once
I Merge additions into computation of H
I Compute H =̂ (h0,h1,h2,h3) = H + (l2, l3)

I Note that H cannot “overflow”
l0 l1 h0 h1 h2 h3

- m̂0 m̂1 m̂2 m̂3

+ l0 l1
+ h0 h1 h2 h3

I Consequence: fewer additions, easier register allocation

Putting it together

Arithmetic cost of n-byte Karatsuba on AVR
I Cost of computing L, M , and H

I 4k + 2 SUB/SBC, 2k EOR for absolute differences
I n + 1 ADD/ADC to add (l0, . . . , lk−1,hk, . . . ,hn−1)

I One EOR to compute t

I A BRNE instruction to branch, then either

I n+ 2 SUB/SBC instructions and one RJMP, or
I n+ 1 ADD/ADC, one CLR, and one NOP

I k ADD/ADC instructions to ripple carry to the end

48-bit Karatsuba on AVR

CLR R22
CLR R23
MOVW R12, R22
MOVW R20, R22

LD R2, X+
LD R3, X+
LD R4, X+
LDD R5, Y+0
LDD R6, Y+1
LDD R7, Y+2

MUL R2, R7
MOVW R10, R0
MUL R2, R5
MOVW R8, R0
MUL R2, R6
ADD R9, R0
ADC R10, R1
ADC R11, R23

MUL R3, R7
MOVW R14, R0
MUL R3, R5
ADD R9, R0
ADC R10, R1
ADC R11, R14
ADC R15, R23
MUL R3, R6
ADD R10, R0
ADC R11, R1
ADC R12, R15

MUL R4, R7
MOVW R14, R0
MUL R4, R5
ADD R10, R0
ADC R11, R1
ADC R12, R14
ADC R15, R23
MUL R4, R6
ADD R11, R0
ADC R12, R1
ADC R13, R15
STD Z+0, R8
STD Z+1, R9
STD Z+2, R10

LD R14, X+
LD R15, X+
LD R16, X+
LDD R17, Y+3
LDD R18, Y+4
LDD R19, Y+5

SUB R2, R14
SBC R3, R15
SBC R4, R16
SBC R26, R26

SUB R5, R17
SBC R6, R18
SBC R7, R19
SBC R27, R27

EOR R2, R26
EOR R3, R26
EOR R4, R26
EOR R5, R27
EOR R6, R27
EOR R7, R27

SUB R2, R26
SBC R3, R26
SBC R4, R26
SUB R5, R27
SBC R6, R27
SBC R7, R27

48-bit Karatsuba on AVR

MUL R14, R19
MOVW R24, R0
MUL R14, R17
ADD R11, R0
ADC R12, R1
ADC R13, R24
ADC R25, R23
MUL R14, R18
ADD R12, R0
ADC R13, R1
ADC R20, R25

MUL R15, R19
MOVW R24, R0
MUL R15, R17
ADD R12, R0
ADC R13, R1
ADC R20, R24
ADC R25, R23
MUL R15, R18
ADD R13, R0
ADC R20, R1
ADC R21, R25

MUL R16, R19
MOVW R24, R0
MUL R16, R17
ADD R13, R0
ADC R20, R1
ADC R21, R24
ADC R25, R23
MUL R16, R18
MOVW R18,R22
ADD R20, R0
ADC R21, R1
ADC R22, R25

MUL R2, R7
MOVW R16, R0
MUL R2, R5
MOVW R14, R0
MUL R2, R6
ADD R15, R0
ADC R16, R1
ADC R17, R23

MUL R3, R7
MOVW R24, R0
MUL R3, R5
ADD R15, R0
ADC R16, R1
ADC R17, R24
ADC R25, R23
MUL R3, R6
ADD R16, R0
ADC R17, R1
ADC R18, R25

MUL R4, R7
MOVW R24, R0
MUL R4, R5
ADD R16, R0
ADC R17, R1
ADC R18, R24
ADC R25, R23
MUL R4, R6
ADD R17, R0
ADC R18, R1
ADC R19, R25

48-bit Karatsuba on AVR

ADD R8, R11
ADC R9, R12
ADC R10, R13
ADC R11, R20
ADC R12, R21
ADC R13, R22
ADC R23, R23

EOR R26, R27
BRNE add_M

SUB R8, R14
SBC R9, R15
SBC R10, R16
SBC R11, R17
SBC R12, R18
SBC R13, R19
SBCI R23, 0
SBC R24, R24
RJMP final

add_M:
ADD R8, R14
ADC R9, R15
ADC R10, R16
ADC R11, R17
ADC R12, R18
ADC R13, R19
CLR R24
ADC R23, R24
NOP

final:
STD Z+3, R8
STD Z+4, R9
STD Z+5, R10
STD Z+6, R11
STD Z+7, R12
STD Z+8, R13

ADD R20, R23
ADC R21, R24
ADC R22, R24

STD Z+9, R20
STD Z+10, R21
STD Z+11, R22

Larger Karatsuba multiplication

I 48-bit Karatsuba is friendly; everything fits into registers
I Remember that previous speed records were achieved by eliminating

loads/stores

I Karatsuba structure needs additional temporary storage
I Good performance needs careful scheduling and register allocation
I Very important is to compute H = H + (lk+1, . . . , ln−1) on the fly
I Use 1-level Karatsuba for 48-bit, 64-bit, 80-bit, 96-bit inputs
I Use 2-level Karatsuba for 128-bit, 160-bit, 192-bit inputs
I Use 3-level Karatsuba for 256-bit inputs

Results

Cycle counts for n-bit multiplication

Input size n
Approach 48 64 80 96 128 160 192 256
Product scanning: 235 395 595 836 — — — —
Hutter, Wenger, 2011: — — — — — 2393 3467 6121
Seo, Kim, 2012: — — — — 1532 2356 3464 6180
Seo, Kim, 2013: — — — — 1523 2341 3437 6115
Karatsuba: 217 360 522 780 1325 1976 2923 4797
— w/o branches: 222 368 533 800 1369 2030 2987 4961

I 160-bit multiplication now > 18% faster
I 256-bit multiplication now > 23% faster

Resources online

Paper:
Michael Hutter, Peter Schwabe. “Multiprecision multiplication on AVR
revisited” .
http://cryptojedi.org/papers/#avrmul

Software: http://cryptojedi.org/crypto/#avrmul

Code generator for operand-scanning, product-scanning, hybrid,
and operand-caching by Hutter and Wenger:
http://mhutter.org/research/avr/index.htm#mulopcache

From 8-bit to 64-bit processors

Main differences (for us)

I Arithmetic on larger (64-bit) integers

I Arithmetic on floating-point numbers
I Arithmetic on vectors
I Pipelined and superscalar execution

Radix-264 representation

I Let’s consider representing 255-bit integers
I Obvious choice: use 4 64-bit integers a0, a1, a2, a3 with

A =

3∑
i=0

ai2
64i

I Arithmetic works just as before (except with larger registers)

Radix-251 representation
I Radix-264 representation works and is sometimes a good choice
I Highly depends on the efficiency of handling carries

I Example 1: Intel Nehalem can do 3 additions every cycle, but only 1
addition with carry every two cycles (carries cost a factor of 6!)

I Example 2: When using vector arithmetic, carries are typically lost
(very expensive to recompute)

I Let’s get rid of the carries, represent A as (a0, a1, a2, a3, a4) with

A =
4∑

i=0

ai2
51·i

I This is called radix-251 representation
I Multiple ways to write the same integer A, for example A = 252:

I (252, 0, 0, 0, 0)
I (0, 2, 0, 0, 0)

I Let’s call a representation (a0, a1, a2, a3, a4) reduced, if all
ai ∈ [0, . . . , 252 − 1]

Addition of two bigint255

typedef struct{
unsigned long long a[5];

} bigint255;

void bigint255_add(bigint255 *r,
const bigint255 *x,
const bigint255 *y)

{
r->a[0] = x->a[0] + y->a[0];
r->a[1] = x->a[1] + y->a[1];
r->a[2] = x->a[2] + y->a[2];
r->a[3] = x->a[3] + y->a[3];
r->a[4] = x->a[4] + y->a[4];

}

I This definitely works for reduced inputs
I This actually works as long as all coefficients are in [0, . . . , 263 − 1]
I We can do quite a few additions before we have to carry (reduce)

Subtraction of two bigint255

typedef struct{
signed long long a[5];

} bigint255;

void bigint255_sub(bigint255 *r,
const bigint255 *x,
const bigint255 *y)

{
r->a[0] = x->a[0] - y->a[0];
r->a[1] = x->a[1] - y->a[1];
r->a[2] = x->a[2] - y->a[2];
r->a[3] = x->a[3] - y->a[3];
r->a[4] = x->a[4] - y->a[4];

}

I Slightly update our bigint255 definition to work with signed 64-bit
integers

I Reduced if coefficients are in [−252 + 1, 252 − 1]

Carrying in radix-251

I With many additions, coefficients may grow larger than 63 bits
I They grow even faster with multiplication

I Eventually we have to carry en bloc:
signed long long carry = r.a[0] >> 51;
r.a[1] += carry;
carry <<= 51;
r.a[0] -= carry;

Big integers and polynomials

I Note: Addition code would look exactly the same for 5-coefficient
polynomial addition

I This is no coincidence: We actually perform arithmetic in Z[x]

I Inputs to addition are 5-coefficient polynomials
I Nice thing about arithmetic Z[x]: no carries!
I To go from Z[x] to Z, evaluate at the radix (this is a ring

homomorphism)
I Carrying means evaluating at the radix
I Thinking of multiprecision integers as polynomials is very powerful

for efficient arithmetic

Using floating-point limbs
I On some microarchitectures floating-point arithmetic is much faster

than integer arithmetic
I An IEEE-754 floating-point number has value

(−1)s · (1.bm−1bm−2 . . . b0) · 2e−t with bi ∈ {0, 1}

I For double-precision floats:
I s ∈ {0, 1} “sign bit”
I m = 52 “mantissa bits”
I e ∈ {1, . . . , 2046} “exponent”
I t = 1023

I For single-precision floats:
I s ∈ {0, 1} “sign bit”
I m = 23 “mantissa bits”
I e ∈ {1, . . . , 254} “exponent”
I t = 127

I Exponent = 0 used to represent 0

I Any number that can be represented like this, will be precise
I Other numbers will be rounded, according to a rounding mode

Addition and subtraction
typedef struct{

double a[12];
} bigint255;

void bigint255_add(bigint255 *r,
const bigint255 *x,
const bigint255 *y)

{
int i;
for(i=0;i<12;i++)

r->a[i] = x->a[i] + y->a[i];
}

void bigint255_sub(bigint255 *r,
const bigint255 *x,
const bigint255 *y)

{
int i;
for(i=0;i<12;i++)

r->a[i] = x->a[i] - y->a[i];
}

Carrying

I For carrying integers we used a right shift (discard lowest bits)

I For floating-point numbers we can use multiplication by the inverse
of the radix

I Example: Radix 222, multiply by 2−22

I This does not cut off lowest bits, need to round
I Some processors have efficient rounding instructions, e.g., vroundpd
I Otherwise (for double-precision):

I add constant 252 + 251

I subtract constant 252 + 251

I This will round the number to an integer according to the rounding
mode (to nearest, towards zero, away from zero, or truncate)

Vector arithmetic

I Most modern processors have vector units
I Work on vectors of floats or ints instead of scalars
I Much larger computational power, e.g., Intel Nehalem:

I 32-bit load throughput: 1 per cycle
I 32-bit add throughput: 3 per cycle
I 32-bit store throughput: 1 per cycle
I 128-bit load throughput: 1 per cycle
I 4× 32-bit add throughput: 2 per cycle
I 128-bit store throughput: 1 per cycle

I Vector instructions are almost as fast as scalar instructions but
do 4× the work

I Some things are cost extra:
I Variably indexed loads (lookups) into vectors
I Data-dependent branches are expensive in SIMD
I Shuffeling entries in vectors

Vectorizing EC scalar multiplication

Computing multiple scalar multiplications
I Changes the rules of the game
I Increases size of active data set

Parallelism inside multiprecision arithmetic
I Addition (in redundant representation) is trivially vectorized
I Vectorizing multiplication needs many shuffles
I Vectorization “eats up” instruction-level parallelism

Parallelism inside EC arithmetic
I Vectorize independent multiplications in EC addition
I May still need some shuffles (after each block of operations)
I Efficiency depends on EC formulas

Example: Montgomery ladder

function ladderstep(xQ−P , XP , ZP , XQ, ZQ)
t1 ← XP + ZP

t6 ← t21
t2 ← XP − ZP

t7 ← t22
t5 ← t6 − t7
t3 ← XQ + ZQ

t4 ← XQ − ZQ

t8 ← t4 · t1
t9 ← t3 · t2
XP+Q ← (t8 + t9)2

ZP+Q ← xQ−P · (t8 − t9)2

X[2]P ← t6 · t7
Z[2]P ← t5 · (t7 + ((A + 2)/4) · t5)
return (X[2]P , Z[2]P , XP+Q, ZP+Q)

end function

Example: Montgomery ladder

function ladderstep(xQ−P , XP , ZP , XQ, ZQ)

t1 ← XP + ZP ; t2 ← XP − ZP ; t3 ← XQ + ZQ; t4 ← XQ − ZQ

t6 ← t1 · t1; t7 ← t2 · t2; t8 ← t4 · t1; t9 ← t3 · t2

t10 ← ((A + 2)/4) · t6
t11 ← ((A + 2)/4− 1) · t7

t5 ← t6 − t7; t4 ← t10 − t11; t1 ← t8 − t9; t0 ← t8 + t9

Z[2]P ← t5 · t4;XP+Q ← t20;X[2]P ← t6 · t7; t2 ← t1 · t1

ZP+Q ← xQ−P · t2

return (X[2]P , Z[2]P , XP+Q, ZP+Q)
end function

A better candidate: Kummer surfaces

I Think of a Kummer surface as the Jacobian of a hyperelliptic curve
modulo negation

I Easier way to think about it:
I Group modulo negation
I Map from group to Kummer surface by rational map X
I Elements represented projectively as (x : y : z : t)
I (x : y : z : t) = (rx : ry : rz : rt) for any r 6= 0
I Efficient doubling and efficient differential addition

I Ladderstep: gets as input X(P ) = (x2 : y2 : z2 : t2),
X(Q) = (x3 : y3 : z3 : t3), and X(Q− P ) = (x1 : y1 : z1 : t1)

I Computes X(2P ) = (x4 : y4 : z4 : t4)
I Computes X(P +Q) = (x5 : y5 : z5 : t5)

I Coordinates are elements of a (large) finite field
I Efficient Kummer surface arithmetic wants efficient multiprecision

arithmetic

Arithmetic on the Kummer surface

Arithmetic on the Kummer surface

The “squared Kummer surface”

I In fact, we use arithmetic on a different, “squared” surface
I Each point (x : y : z : t) on the original surface corresponds to

(x2 : y2 : z2 : t2) on the squared surface
I No operation-count advantages
I Easier to construct squared surface with small constants
I In the following rename (x2 : y2 : z2 : t2) to (x : y : z : t)

Arithmetic on the squared Kummer surface

Arithmetic on the squared Kummer surface

Arithmetic on the (original) Kummer surface

A suitable Kummer surface

I Formulas for efficient Kummer surface arithmetic known for a while
I Originally proposed by Chudnovsky, Chudnovsky, 1986
I 10M+ 9S+ 6m formulas by Gaudry, 2006
I 7M+ 12S+ 9m formulas by Bernstein, 2006

I Problem: find cryptographically secure surface with small constants
I Gaudry, Schost, 2012: suitable (squared) surface:

I Defined over the field F2127−1

I (1 : a2/b2 : a2/c2 : a2/d2) = (−114 : 57 : 66 : 418)
I (1 : A2/B2 : A2/C2 : A2/D2) = (−833 : 2499 : 1617 : 561)

I Finding this surface cost 1000000 CPU hours
I The same surface has been used by Bos, Costello, Hisil, and Lauter

(Eurocrypt 2013)

Sandy Bridge/Ivy Bridge

I Pre-latest microarchitectures by Intel
I Very powerful vector instructions: AVX
I Operations on 256-bit vector registers

I Can only do vectors of (single- or double-precision) floats
I Performance:

I 1 vectorized double-precision multiplication per cycle, plus
I 1 vectorized double-precision addition per cycle
I A total of 8 double-precision operations per cycle!

Representing elements of F2127−1

I Represent an element A in radix-2127/6

I Write A as a0, a1, a2, a3, a4, a5, where
I a0 is a small multiple of 20
I a1 is a small multiple of 222
I a2 is a small multiple of 243
I a3 is a small multiple of 264
I a4 is a small multiple of 285
I a5 is a small multiple of 2106

Multiplication

I Consider multiplication of A and B with reduction mod 2127 − 1

I Make use of the fact that 2127 ≡ 1

I With radix 2127/6 we obtain:

r0 = a0b0 + 2−127a1b5 + 2−127a2b4 + 2−127a3b3 + 2−127a4b2 + 2−127a5b1

r1 = a0b1 + a1b0 + 2−127a2b5 + 2−127a3b4 + 2−127a4b3 + 2−127a5b2

r2 = a0b2 + a1b1 + a2b0 + 2−127a3b5 + 2−127a4b4 + 2−127a5b3

r3 = a0b3 + a1b2 + a2b1 + a3b0 + 2−127a4b5 + 2−127a5b4

r4 = a0b4 + a1b3 + a2b2 + a3b1 + a4b0 + 2−127a5b5

r5 = a0b5 + a1b4 + a2b3 + a3b2 + a4b1 + a5b0

I Obviously, we always perform this whole thing 4× in parallel
I Obviously, we specialize squaring
I Obviously, we specialize multiplications by small constants

The Hadamard transform

I Only shuffeling operation in
Kummer arithmetic

I AVX has limited shuffeling
across left and right half

I Plain Hadamard turns out to be
expensive

Permuted and negated Hadamard
I Allow generalized Hadamard to output permuted vector
I Self-inverting permutation “cleans” after two generalized Hadamards

I Allow generalized Hadamard to negate vector entries
I “Clean” negations by multiplication by negated constants

Arithmetic on the squared Kummer surface

Results

128-bit secure, constant-time scalar multiplication

arch cycles open g source of software
Sandy 194036 yes 1 Bernstein–Duif–Lange–Schwabe–

Yang CHES 2011
Sandy 153000? no 1 Hamburg
Sandy 137000? no 1 Longa–Sica Asiacrypt 2012
Sandy 122716 yes 2 Bos–Costello–Hisil–Lauter Euro-

crypt 2013
Sandy 119904 yes 1 Oliveira–López–Aranha–Rodríguez-

Henríquez CHES 2013
Sandy 96000? no 1 Faz-Hernández–Longa–Sánchez CT-

RSA 2014
Sandy 92000? no 1 Faz-Hernández–Longa–Sánchez

July 2014
Sandy 88916 yes 2 new (our results)

Results

128-bit secure, constant-time scalar multiplication

arch cycles open g source of software
Ivy 182708 yes 1 Bernstein–Duif–Lange–Schwabe–Yang

CHES 2011
Ivy 145000? yes 1 Costello–Hisil–Smith Eurocrypt 2014
Ivy 119032 yes 2 Bos–Costello–Hisil–Lauter Euro-

crypt 2013
Ivy 114036 yes 1 Oliveira–López–Aranha–Rodríguez-

Henríquez CHES 2013
Ivy 92000? no 1 Faz-Hernández–Longa–Sánchez CT-

RSA 2014
Ivy 89000? no 1 Faz-Hernández–Longa–Sánchez

July 2014
Ivy 88448 yes 2 new (our results)

More results

Also optimized for Intel Haswell

arch cycles open g source of software
Haswell 145907 yes 1 Bernstein–Duif–Lange–

Schwabe–Yang CHES 2011
Haswell 100895 yes 2 Bos–Costello–Hisil–Lauter

Eurocrypt 2013
Haswell 55595 no 1 Oliveira–López–Aranha–

Rodríguez-Henríquez
CHES 2013

Haswell 54389 yes 2 new (our results)

Even more results

Also optimized for ARM Cortex-A8

arch cycles open g source of software
A8-slow 497389 yes 1 Bernstein–Schwabe CHES 2012
A8-slow 305395 yes 2 new (our result)
A8-fast 460200 yes 1 Bernstein–Schwabe CHES 2012
A8-fast 273349 yes 2 new (our result)

Resources online

Paper:
Daniel J. Bernstein, Chitchanok Chuengsatiansup, Tanja Lange, Peter
Schwabe. “Kummer strikes back: new DH speed records” .
To be presented at Asiacrypt 2014
http://cryptojedi.org/papers/#kummer

Software:
Included in SUPERCOP, subdirectory crypto_scalarmult/kummer/
http://bench.cr.yp.to/supercop.html

